
************************************************************1
*                                                          *2
*            FinderLaunch - An AppleWorks Init             *3
*                                                          *4
*       - Prevents Crash to Monitor when AppleWorks        *5
*         is Re-launched from the GS/OS Finder after       *6
*           previously launching and quitting in           *7
*                     a single session.                    *8
*                                                          *9
*       - Init must be present in the AppleWorks           *10
*                   inits directory                        *11
*             (e.g. /HD/AppleWorks/AW.Inits)               *12
*       - Init Manager must be activated in                * 13
*          AppleWorks 'Standard Settings' Menu             *14
*                                                          * 15
*          Version 1.0 (for AppleWorks Version 5.1)        *16
*                    (c) 2015 Hugh Hood                    *17
*                                                          *18
************************************************************19
20
 TR ADR ; truncate bank address21
22
 XC  ; enable 65C02 code23
24
* Equates *25
26
EMULSTACK EQU $100 ; in AuxMEM - $01/010027
Seg00Type EQU $FE0 ; memory manager type28
   ;  $41 = SEG.AM / $52 = SEG.RM (IIGS) /29
   ;  $58 = SEG.XM / $4D = Deja IIx (Mac)30
InitAdr EQU $4000 ; load address for Init files31
MainZP EQU $C008 ; Main Zero Page (AppleWorks)32
AuxZP EQU $C009 ; Aux Zero Page (UltraMacros)33
34
35
36
 ORG InitAdr ; ($4000)37
38
********************************39
*          Init Header         *40
********************************41
START42
 JMP IStart ; skip over header43
44
**--------------------------------------------------------------45
46
 ASC 'mb' ; Init ID Bytes (AW 5.1)47
 DB $0A ; Init Version - programmer assigned48
   ;  e.g. - $0A/1.0 $0B/1.1 $1A/2.649
 STR 'FinderLaunch'; Init Screen Name (max 18 character)50
 HEX 0000 ; Header End Bytes51
52
**--------------------------------53
54
55
IStart LDA Seg00Type ; Memory Manager Type56



 CMP #$52 ; Is it Apple IIGS? (SEG.RM)57
 BNE Done ; disregard - IIGS only58
59
 SEI  ; disable interrupts60
 LDA #$7F ; change value to start halfway61
 STA AuxZP ; switch in AuxZP and LC62
   ;  (UltraMacros runs in AuxMEM)63
 STA EMULSTACK ; store MainZP Stack pointer here 64
   ;  when AuxZP and LC are in use65
66
67
 STA MainZP ; switch in MainZP and LC68
   ;  (AppleWorks runs in MainMEM)69
 CLI  ; re-enable interrupts70
71
*-----------------------------------------------------------72
*73
*   NOTE: For background information on this issue, see page 152-153 of the74
*          Apple IIe Technical Reference Manual, page 13 of the Apple IIgs75
*          Firmware Reference Manual, IIgs Tech Note #88, GS/OS source code76
*          for 'GQuit', and GSBug revision notes for Version 1.5b15.77
*78
* After quitting AppleWorks and returning to the GS/OS Finder, the location,79
* $01/100, also called EMULSTACK, [and 'mnemstkptr' in the GS/OS GQuit/P8Quit80
* source code] is set to the value of the stack pointer, which will contain81
* either a $B2 (AW 5.1) or a $B4 (AW 3.0 with Ultra4).82
*83
* This value is NOT placed there by AppleWorks, but by GS/OS, which is using84
* a routine located at $E0/D8AD in language card Bank 0 memory. This routine85
* reads the value of the page 1 stack pointer and saves it at EMULSTACK.86
*87
* By convention, EMULSTACK is SUPPOSED to receive the value of MAIN zero page88
* stack pointer prior to a program's switching in the AUXMEM zero page and89
* stack. This allows interrupting programs (and certain IIGS tools), when90
* called while the AUXMEM zero page and stack are active, to set the correct91
* stack pointer prior to switching back to the MAIN zero page stack to92
* accomplish their task(s). AppleWorks, however, does not follow this93
* convention when switching between the MAIN and AUXMEM zero page and stack,94
* even though UltraMacros, when enabled, makes use of the AUXMEM zero page95
* and stack, and obviously must have some other method of preserving the 96
* respective stack pointers prior to switching back and forth.97
*                (Probably MainStackPtr @ $B4FC in Main Memory)98
*99
* Consequently, upon a SECOND launch of AppleWorks an interrupting program or100
* tool will find an INCORRECT value ($B2 or $B4) in EMULSTACK, but will assume101
* that it is in fact a valid stack pointer for the MAIN stack, and will make102
* use of the stack from $01B2/B4 - $0100 for servicing the interrupt.103
*104
* Since it is not valid, and very likely to be too 'high' on the stack, the105
* end result is that the MAIN zero page stack is corrupted when AppleWorks106
* regains control. AppleWorks generally crashes at this point.107
* 108
* By setting EMULSTACK to $7F prior to starting AppleWorks, we can ensure109
* that any interrupting routines will ONLY use stack space between $0100-$017F,110
* and leaving $0180-$01FF for AppleWorks' exclusive use.111
*112



* Credit for this 'technique' belongs to Glen Bredon, who implemented it in113
* his PS.16.TO.8 from ProSEL 16 as a pre-launch to ANY ProDOS 8 program 114
* launched from GS/OS.115
*116
*-----------------------------------------------------------117
118
Done RTS  ; back to Init Manager119
120
PatchEnd EQU *121
 SAV I.FINDERLAUNCH122
 LST OFF123
124
 END125

126


